Paired t-test Using SAS

(commands= finan_paired.sas)
This command file uses the data set, PULSE, which is imported from the PULSE.XLS data set, using Proc Import.
/**************************************

 PAIRED T-TEST

FILENAME: DAY2_FINAN_PAIRED.SAS

****************************************/
options nodate nofmterr pagno=1;

title;

options FORMCHAR="|----|+|---+=|-/\<>*";

/*IMPORT PULSE DATA USING PROC IMPORT*/

PROC IMPORT OUT= WORK.PULSE 

            DATAFILE= "C:\temp\labdata\PULSE.XLS" 

            DBMS=EXCEL REPLACE;

     SHEET="pulse"; 

     GETNAMES=YES;

     MIXED=NO;

     SCANTEXT=YES;

     USEDATE=YES;

     SCANTIME=YES;

RUN;

proc format lib=work;

  value ranfmt 1="Ran"

               2="Didn't Run";

  value sexfmt 1="Female"

               0="Male";

  value minfmt 1="Minority"

               0="White";

  value actfmt 1="Low Activity"

               2="Medium Activity"



    3="High Activity";

run;

/*PROC CONTENTS TO CHECK DATA*/

title "Pulse Data Set";

proc contents data=pulse varnum;

run;
                                      Pulse Data Set                                       

                                    The CONTENTS Procedure

     Data Set Name        WORK.PULSE                             Observations          92

     Member Type          DATA                                   Variables             8

     Engine               V9                                     Indexes               0

     Created              Monday, August 21, 2006 09:27:51 PM    Observation Length    64

     Last Modified        Monday, August 21, 2006 09:27:51 PM    Deleted Observations  0

     Protection                                                  Compressed            NO

     Data Set Type                                               Sorted                NO

     Label

     Data Representation  WINDOWS_32

     Encoding             wlatin1  Western (Windows)

                                  Variables in Creation Order

                           #    Variable    Type    Len    Label

                           1    pulse1      Num       8    pulse1

                           2    pulse2      Num       8    pulse2

                           3    ran         Num       8    ran

                           4    smokes      Num       8    smokes

                           5    sex         Num       8    sex

                           6    height      Num       8    height

                           7    weight      Num       8    weight

                           8    activity    Num       8    activity

/*CALCULATE PULSEDIFF VARIABLE IN A DATA STEP*/

data newpulse;

  set pulse;

  pulsediff = pulse2 - pulse1;

run;

/*CHECK NORMALITY OF PULSE DIFFERENCE FOR LEVELS OF  RAN*/

title "Check distribution of PULSEDIFF, and one-sample t-test";

proc univariate data=newpulse normal;

class ran;

format ran ranfmt.;

var pulsediff;

run;
                          Check distribution of PULSEDIFF, and one-sample t-test                   

                                   The UNIVARIATE Procedure

                                     Variable:  pulsediff

                                       ran = Ran

                                  Tests for Location: Mu0=0

                       Test           -Statistic-    -----p Value------

                       Student's t    t  7.435276    Pr > |t|    <.0001

                       Sign           M        15    Pr >= |M|   <.0001

                       Signed Rank    S       286    Pr >= |S|   <.0001

                                     Tests for Normality

                  Test                  --Statistic---    -----p Value------

                  Shapiro-Wilk          W     0.950493    Pr < W      0.1173

                  Kolmogorov-Smirnov    D     0.148201    Pr > D      0.0493

                  Cramer-von Mises      W-Sq   0.11509    Pr > W-Sq   0.0711

                  Anderson-Darling      A-Sq  0.678521    Pr > A-Sq   0.0738

                                   The UNIVARIATE Procedure

                                     Variable:  pulsediff

                                       ran = Didn't Run

                                  Tests for Location: Mu0=0

                       Test           -Statistic-    -----p Value------

                       Student's t    t  -0.19101    Pr > |t|    0.8492

                       Sign           M      -2.5    Pr >= |M|   0.5327

                       Signed Rank    S       -30    Pr >= |S|   0.6979

                                    Tests for Normality

                  Test                  --Statistic---    -----p Value------

                  Shapiro-Wilk          W     0.931434    Pr < W      0.0031

                  Kolmogorov-Smirnov    D     0.174118    Pr > D     <0.0100

                  Cramer-von Mises      W-Sq  0.265908    Pr > W-Sq  <0.0050

                  Anderson-Darling      A-Sq  1.475438    Pr > A-Sq  <0.0050

/*PAIRED t-test OF PULSE2 vs. PULSE1*/

title "Paired t-test of Pulse2 vs. Pulse1";

proc ttest data=newpulse;

  paired pulse2*pulse1;

run;
                              Paired t-test of Pulse2 vs. Pulse1                             

                                      The TTEST Procedure

                                          Statistics

                          Lower CL          Upper CL  Lower CL           Upper CL

  Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

  pulse2 - pulse1     92    4.3406  7.1304    9.9203    11.766   13.471    15.759   1.4045

                                           T-Tests

                        Difference           DF    t Value    Pr > |t|

                        pulse2 - pulse1      91       5.08      <.0001

/*REDO PAIRED t-test FOR EACH LEVEL OF RAN*/

proc sort data=newpulse;

by ran;

run;

title "Paired t-test of Pulse2 vs Pulse1 by RAN";

proc ttest data=newpulse;

  by ran;

  format ran ranfmt.;

  paired pulse2*pulse1;

run;

                              Paired t-test of Pulse2 vs Pulse1 by RAN                          

------------------------------------------- ran=Ran ------------------------------------------
                                      The TTEST Procedure

                                          Statistics

                          Lower CL          Upper CL  Lower CL           Upper CL

  Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

  pulse2 - pulse1     35    13.745  18.914    24.084    12.173    15.05    19.718   2.5439

                                          T-Tests

                        Difference           DF    t Value    Pr > |t|

                        pulse2 - pulse1      34       7.44      <.0001

--------------------------------------- ran=Didn't Run ----------------------------------------

                                      The TTEST Procedure

                                          Statistics

                          Lower CL          Upper CL  Lower CL           Upper CL

  Difference           N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

  pulse2 - pulse1     57    -1.209  -0.105    0.9987    3.5126   4.1605    5.1039   0.5511

                                           T-Tests
                        Difference           DF    t Value    Pr > |t|
                        pulse2 - pulse1      56      -0.19      0.8492

proc sort data=newpulse;

 by ran;

run;

goptions reset=all;

goptions device=win target=winprtm;

title "Boxplot";

proc boxplot data=newpulse;

  plot pulsediff * ran / boxstyle=schematic;

  format ran ranfmt.;

run;

title "Independent Samples t-test on Pulsediff";

proc ttest data=newpulse;

  class ran;

  format ran ranfmt.;

  var pulsediff;

run;
                                     The TTEST Procedure

                                          Statistics

                             Lower CL          Upper CL  Lower CL           Upper CL

Variable  ran             N      Mean    Mean      Mean   Std Dev  Std Dev   Std Dev  Std Err

pulsediff  Ran            35    13.745  18.914    24.084    12.173    15.05    19.718   2.5439

pulsediff  Didn't Run     57    -1.209  -0.105    0.9987    3.5126   4.1605    5.1039   0.5511

pulsediff  Diff (1-2)           14.832   19.02    23.207    8.5669    9.815    11.492   2.1077

                                           T-Tests

            Variable     Method           Variances      DF    t Value    Pr > |t|

            pulsediff    Pooled           Equal          90       9.02      <.0001

            pulsediff    Satterthwaite    Unequal      37.2       7.31      <.0001

                                    Equality of Variances

                Variable     Method      Num DF    Den DF    F Value    Pr > F

                pulsediff    Folded F        34        56      13.08    <.0001

/*NON-PARAMETRIC WILCOXON RANK SUM TEST*/
title "Wilcoxon Rank Sum test on Pulsediff";

proc npar1way data=newpulse wilcoxon;

  class ran;

  format ran ranfmt.;

  var pulsediff;

run;
                              Wilcoxon Rank Sum test on Pulsediff                            

                                    The NPAR1WAY Procedure

                      Wilcoxon Scores (Rank Sums) for Variable pulsediff

                                  Classified by Variable ran

                                    Sum of      Expected       Std Dev          Mean

          ran              N        Scores      Under H0      Under H0         Score

          --------------------------------------------------------------------------

          Ran             35       2440.50       1627.50    123.660110     69.728571

          Didn't Run      57       1837.50       2650.50    123.660110     32.236842

                              Average scores were used for ties.
                                   Wilcoxon Two-Sample Test

                                Statistic             2440.5000
                                Normal Approximation

                                Z                        6.5704

                                One-Sided Pr >  Z        <.0001

                                Two-Sided Pr > |Z|       <.0001

                                t Approximation

                                One-Sided Pr >  Z        <.0001

                                Two-Sided Pr > |Z|       <.0001
                           Z includes a continuity correction of 0.5.

                                      Kruskal-Wallis Test
                                Chi-Square              43.2237

                                DF                            1

                                Pr > Chi-Square          <.0001

